Task 2 — DAX Measures Using Variables

A set of 10 DAX measures built on the `Fact_Employee` star schema model, each written using the `VAR` / `RETURN` pattern to support business decision-making around employee attrition.

Quick Reference

#	Measure	Result	Purpose
1	Total Employees	1,470	Baseline count for all ratios
2	Attrition Count	237	Raw count of employees who left
3	Attrition Rate %	16.1%	Core KPI — share of workforce lost
4	Avg Income Active	$6,833	Benchmark income for retained staff
5	Avg Income Leavers	$4,787	Income benchmark for leavers
6	Income Gap %	29.9%	Pay gap between leavers and active staff
7	Avg Tenure Leavers	5.1 years	Typical tenure at time of leaving
8	Attrition Rate OverTime	30.5%	Attrition rate among overtime workers
9	Attrition Rate Variance	Dynamic	Flags above/below-average risk by category
10	Avg Years Since Promotion Leavers	Dynamic	Promotion stagnation among leavers

---

Measure Definitions

1. Total Employees

```
Total Employees =
VAR EmployeeCount = COUNTROWS(Fact_Employee)
RETURN
    EmployeeCount
```

Result: 1,470 — Baseline count for all other ratios.

---

2. Attrition Count

```
Attrition Count =
VAR LeaversCount =
    CALCULATE(COUNTROWS(Fact_Employee), Fact_Employee[Attrition] =
"Yes")
RETURN
    LeaversCount
```

Result: 237

---

3. Attrition Rate %

```
Attrition Rate % =
VAR TotalEmp = COUNTROWS(Fact_Employee)
VAR LeaversCount =
    CALCULATE(COUNTROWS(Fact_Employee), Fact_Employee[Attrition] =
"Yes")
RETURN
    DIVIDE(LeaversCount, TotalEmp, 0)
```

Uses `DIVIDE()` instead of `/` for safe division (returns 0 instead of erroring on an empty filter context). **Result: 16.1%**

---

4. Avg Income Active

```
Avg Income Active =
VAR ActiveIncome =
    CALCULATE(AVERAGE(Fact_Employee[MonthlyIncome]),
Fact_Employee[Attrition] = "No")
RETURN
    ActiveIncome
```

Result: $6,833

---

5. Avg Income Leavers

```
Avg Income Leavers =
VAR LeaversIncome =
    CALCULATE(AVERAGE(Fact_Employee[MonthlyIncome]),
Fact_Employee[Attrition] = "Yes")
RETURN
    LeaversIncome
```

Result: $4,787

---

6. Income Gap %

```
Income Gap % =
VAR ActiveIncome =
    CALCULATE(AVERAGE(Fact_Employee[MonthlyIncome]),
Fact_Employee[Attrition] = "No")
VAR LeaversIncome =
    CALCULATE(AVERAGE(Fact_Employee[MonthlyIncome]),
Fact_Employee[Attrition] = "Yes")
RETURN
    DIVIDE(ActiveIncome - LeaversIncome, ActiveIncome, 0)
```

Result: 29.9%

---

7. Avg Tenure Leavers

```
Avg Tenure Leavers =
VAR LeaversTenure =
    CALCULATE(AVERAGE(Fact_Employee[YearsAtCompany]),
Fact_Employee[Attrition] = "Yes")
RETURN
    LeaversTenure
```

Result: 5.1 years

---

8. Attrition Rate OverTime

```
Attrition Rate OverTime =
VAR OvertimeEmployees =
    CALCULATE(COUNTROWS(Fact_Employee), Fact_Employee[OverTimeID] =
2)
VAR OvertimeLeavers =
    CALCULATE(COUNTROWS(Fact_Employee), Fact_Employee[OverTimeID] =
2, Fact_Employee[Attrition] = "Yes")
RETURN
    DIVIDE(OvertimeLeavers, OvertimeEmployees, 0)
```

`OverTimeID = 2` corresponds to “Yes” (alphabetical dimension build: No = 1, Yes = 2). **Result: 30.5%**

---

9. Attrition Rate Variance

```
Attrition Rate Variance =
VAR CurrentRate =
    DIVIDE(
        CALCULATE(COUNTROWS(Fact_Employee), Fact_Employee[Attrition]
= "Yes"),
        COUNTROWS(Fact_Employee), 0
    )
VAR OverallRate =
    DIVIDE(
        CALCULATE(COUNTROWS(Fact_Employee), Fact_Employee[Attrition]
= "Yes", ALL(Fact_Employee)),
        CALCULATE(COUNTROWS(Fact_Employee), ALL(Fact_Employee)), 0
    )
RETURN
    CurrentRate - OverallRate
```

Compares the rate in the current visual context to the company-wide rate (via `ALL()` to strip filters). Used to flag outliers in charts sliced by Department/JobRole — any large positive value signals above-average risk. **Result: Dynamic** (context-dependent)

---

10. Avg Years Since Promotion Leavers

```
Avg Years Since Promotion Leavers =
VAR LeaversPromotionGap =
    CALCULATE(AVERAGE(Fact_Employee[YearsSinceLastPromotion]),
Fact_Employee[Attrition] = "Yes")
RETURN
    LeaversPromotionGap
```

Result: Dynamic (context-dependent)

Validation

All 10 measures were tested in a Table visual in Power BI and cross-checked against an independent Python/pandas calculation on the same dataset — all values matched exactly.